

A decorative graphic on the left side of the slide, consisting of a network of light blue lines and small circles, resembling a circuit board or a stylized tree structure, set against a dark blue gradient background.

QUIZ UVM

EL DUT

- Este es un detector de patrones simple escrito en Verilog para identificar un patrón en un flujo de valores de entrada.
- En cada reloj, hay una nueva entrada al diseño y cuando coincide con el patrón "1011", la salida (out) se configurará en 1. Para este propósito, el diseño se implementa como una máquina de estado que se mueve a través de diferentes etapas. a medida que avanza a través de la secuencia de identificación de patrones.

PRUEBA BÁSICA

El banco de pruebas de verificación se desarrollará en UVM:

- La secuencia genera un flujo aleatorio de valores de entrada que se pasarán al controlador como `uvm_sequence_item`
- El driver recibe el artículo y lo conduce al DUT a través de una interfaz virtual
- El monitor captura valores en el pin de entrada y salida del DUT, crea un paquete y lo envía al scoreboard
- El cuadro de indicadores es el principal responsable de verificar la corrección funcional del diseño en función de los valores de entrada y salida que recibe del monitor.

El flujo de valores de entrada debe ser aleatorio para lograr la máxima eficiencia. Debería poder captar los siguientes escenarios:

- 011011011010
- 101011100
- 111011011

PARA CORRER UNA PRUEBA QUE USA UVM EN VCS

Añada el switch:

- `-ntb_opts uvm-1.2`


```

1 module det_1011 ( input clk,
2                   input rstn,
3                   input in,
4                   output out );
5
6     parameter IDLE    = 0,
7                   S1    = 1,
8                   S10    = 2,
9                   S101    = 3,
10                  S1011    = 4;
11
12     reg [2:0] cur_state, next_state;
13
14     assign out = cur_state == S1011 ? 1 : 0;
15
16     always @ (posedge clk) begin
17         if (!rstn)
18             cur_state <= IDLE;
19         else
20             cur_state <= next_state;
21     end
22
23     always @ (cur_state or in) begin
24         case (cur_state)
25             IDLE : begin
26                 if (in) next_state = S1;
27                 else next_state = IDLE;
28             end
29
30             S1: begin
31                 if (in) next_state = S1;
32                 else next_state = S10;
33             end
34
35             S10 : begin
36                 if (in) next_state = S101;
37                 else next_state = IDLE;
38             end

```

```

40         S101 : begin
41             if (in) next_state = S1011;
42             else next_state = S10;
43         end
44
45         S1011: begin
46             if (in) next_state = S1;
47             else next_state = S10;
48         end
49     endcase
50 end
51 endmodule

```

The background is a blue gradient. In the corners, there are decorative white lines resembling circuit traces or a stylized tree structure, with small circles at the end of the lines.

RESPUESTA

SEQUENCE ITEM

```
1 // This is the base transaction object that will be used
2 // in the environment to initiate new transactions and
3 // capture transactions at DUT interface
4 class Item extends uvm_sequence_item;
5     `uvm_object_utils(Item)
6     rand bit in;
7     bit out;
8
9     virtual function string convert2str();
10         return $sformatf("in=%0d, out=%0d", in, out);
11     endfunction
12
13     function new(string name = "Item");
14         super.new(name);
15     endfunction
16
17     constraint c1 { in dist {0:/20, 1:/80}; }
18 endclass
```

SEQUENCE

```
1 class gen_item_seq extends uvm_sequence;
2   `uvm_object_utils(gen_item_seq)
3   function new(string name="gen_item_seq");
4     super.new(name);
5   endfunction
6
7   rand int num;      // Config total number of items to be sent
8
9   constraint c1 { soft num inside {[10:50]}; }
10
11  virtual task body();
12    for (int i = 0; i < num; i++) begin
13      Item m_item = Item::type_id::create("m_item");
14      start_item(m_item);
15      m_item.randomize();
16      `uvm_info("SEQ", $sformatf("Generate new item: %s", m_item.convert2str()), UVM_HIGH)
17      finish_item(m_item);
18    end
19    `uvm_info("SEQ", $sformatf("Done generation of %0d items", num), UVM_LOW)
20  endtask
21 endclass
```


DRIVER

```
1 // The driver is responsible for driving transactions to the DUT
2 // All it does is to get a transaction from the mailbox if it is
3 // available and drive it out into the DUT interface.
4 class driver extends uvm_driver #(Item);
5     `uvm_component_utils(driver)
6     function new(string name = "driver", uvm_component parent=null);
7         super.new(name, parent);
8     endfunction
9
10    virtual des_if vif;
11
12    virtual function void build_phase(uvm_phase phase);
13        super.build_phase(phase);
14        if (!uvm_config_db#(virtual des_if)::get(this, "", "des_vif", vif))
15            `uvm_fatal("DRV", "Could not get vif")
16    endfunction
17
18    virtual task run_phase(uvm_phase phase);
19        super.run_phase(phase);
20        forever begin
21            Item m_item;
22            `uvm_info("DRV", $sformatf("Wait for item from sequencer"), UVM_HIGH)
23            seq_item_port.get_next_item(m_item);
24            drive_item(m_item);
25            seq_item_port.item_done();
26        end
27    endtask
28
29    virtual task drive_item(Item m_item);
30        @(vif.cb);
31        vif.cb.in <= m_item.in;
32    endtask
33 endclass
```

MONITOR

```
22 virtual task run_phase(uvm_phase phase);
23     super.run_phase(phase);
24     // This task monitors the interface for a complete
25     // transaction and writes into analysis port when complete
26     forever begin
27         @ (vif.cb);
28         if (vif.rstn) begin
29             Item item = Item::type_id::create("item");
30             item.in = vif.in;
31             item.out = vif.cb.out;
32             mon_analysis_port.write(item);
33             `uvm_info("MON", $sformatf("Saw item %s", item.convert2str()), UVM_HIGH)
34         end
35     end
36 endtask
37 endclass
```

```
1 // The monitor has a virtual interface handle with which
2 // it can monitor the events happening on the interface.
3 // It sees new transactions and then captures information
4 // into a packet and sends it to the scoreboard
5 // using another mailbox.
6 class monitor extends uvm_monitor;
7     `uvm_component_utils(monitor)
8     function new(string name="monitor", uvm_component parent=null);
9         super.new(name, parent);
10    endfunction
11
12    uvm_analysis_port #(Item) mon_analysis_port;
13    virtual des_if vif;
14
15    virtual function void build_phase(uvm_phase phase);
16        super.build_phase(phase);
17        if (!uvm_config_db#(virtual des_if)::get(this, "", "des_vif", vif))
18            `uvm_fatal("MON", "Could not get vif")
19        mon_analysis_port = new ("mon_analysis_port", this);
20    endfunction
```

SCOREBOARD

```
1 // The scoreboard is responsible to check design functionality and
2 // should track input and try to match the pattern and ensure that
3 // the design has found the pattern as well. The scoreboard should
4 // flag an error if the design didnt find the pattern and ensure
5 // that "out" remains zero, and if the design found the pattern,
6 // "out" is set to the correct value.
7 class scoreboard extends uvm_scoreboard;
8     `uvm_component_utils(scoreboard)
9     function new(string name="scoreboard", uvm_component parent=null);
10         super.new(name, parent);
11     endfunction
12
13     bit[`LENGTH-1:0]    ref_pattern;
14     bit[`LENGTH-1:0]    act_pattern;
15     bit                  exp_out;
16
17     uvm_analysis_imp #(Item, scoreboard) m_analysis_imp;
18
19     virtual function void build_phase(uvm_phase phase);
20         super.build_phase(phase);
21
22         m_analysis_imp = new("m_analysis_imp", this);
23         if (!uvm_config_db#(bit[`LENGTH-1:0])::get(this, "", "ref_pattern", ref_pattern))
24             `uvm_fatal("SCBD", "Did not get ref_pattern !")
25     endfunction
```

```
27 virtual function write(Item item);
28     act_pattern = act_pattern << 1 | item.in;
29
30     `uvm_info("SCBD", $sformatf("in=%0d out=%0d ref=0b%0b act=0b%0b",
31                                 item.in, item.out, ref_pattern, act_pattern), UVM_LOW)
32
33     // Always check that expected out value is the actual observed value
34     // Since it takes 1 clock for out to be updated after pattern match,
35     // do the check first and then update exp_out value
36     if (item.out != exp_out) begin
37         `uvm_error("SCBD", $sformatf("ERROR ! out=%0d exp=%0d",
38                                     item.out, exp_out))
39     end else begin
40         `uvm_info("SCBD", $sformatf("PASS ! out=%0d exp=%0d",
41                                     item.out, exp_out), UVM_HIGH)
42     end
43
44     // If current index has reached the full pattern, then set exp_out to be 1
45     // which will be checked in the next clock. If pattern is not complete, keep
46     // exp_out to zero
47     if (!(ref_pattern ^ act_pattern)) begin
48         `uvm_info("SCBD", $sformatf("Pattern found to match, next out should be 1"), UVM_LOW)
49         exp_out = 1;
50     end else begin
51         exp_out = 0;
52     end
53
54 endfunction
55 endclass
```


AGENT

```
1 // Create an intermediate container called "agent" to hold
2 // driver, monitor and sequencer
3 class agent extends uvm_agent;
4     `uvm_component_utils(agent)
5     function new(string name="agent", uvm_component parent=null);
6         super.new(name, parent);
7     endfunction
8
9     driver      d0;          // Driver handle
10    monitor      m0;          // Monitor handle
11    uvm_sequencer #(Item) s0;      // Sequencer Handle
12
13    virtual function void build_phase(uvm_phase phase);
14        super.build_phase(phase);
15        s0 = uvm_sequencer#(Item)::type_id::create("s0", this);
16        d0 = driver::type_id::create("d0", this);
17        m0 = monitor::type_id::create("m0", this);
18    endfunction
19
20    virtual function void connect_phase(uvm_phase phase);
21        super.connect_phase(phase);
22        d0.seq_item_port.connect(s0.seq_item_export);
23    endfunction
24
25 endclass
```


AMBIENTE

```
26
27 // The environment is a container object simply to hold
28 // all verification components together. This environment can
29 // then be reused later and all components in it would be
30 // automatically connected and available for use
31 class env extends uvm_env;
32     `uvm_component_utils(env)
33     function new(string name="env", uvm_component parent=null);
34         super.new(name, parent);
35     endfunction
36
37     agent      a0;          // Agent handle
38     scoreboard sb0;         // Scoreboard handle
39
40     virtual function void build_phase(uvm_phase phase);
41         super.build_phase(phase);
42         a0 = agent::type_id::create("a0", this);
43         sb0 = scoreboard::type_id::create("sb0", this);
44     endfunction
45
46     virtual function void connect_phase(uvm_phase phase);
47         super.connect_phase(phase);
48         a0.m0.mon_analysis_port.connect(sb0.m_analysis_imp);
49     endfunction
50 endclass
```

PRUEBA

```
1 // Test class instantiates the environment and starts it.
2 class base_test extends uvm_test;
3     `uvm_component_utils(base_test)
4     function new(string name = "base_test", uvm_component parent=null);
5         super.new(name, parent);
6     endfunction
7
8     env            e0;
9     bit[`LENGTH-1:0] pattern = 4'b1011;
10    gen_item_seq    seq;
11    virtual des_if  vif;
12
13    virtual function void build_phase(uvm_phase phase);
14        super.build_phase(phase);
15
16        // Create the environment
17        e0 = env::type_id::create("e0", this);
18
19        // Get virtual IF handle from top level and pass it to everything
20        // in env level
21        if (!uvm_config_db#(virtual des_if)::get(this, "", "des_vif", vif))
22            `uvm_fatal("TEST", "Did not get vif")
23        uvm_config_db#(virtual des_if)::set(this, "e0.a0.*", "des_vif", vif);
24
25        // Setup pattern queue and place into config db
26        uvm_config_db#(bit[`LENGTH-1:0])::set(this, "*", "ref_pattern", pattern);
27
28        // Create sequence and randomize it
29        seq = gen_item_seq::type_id::create("seq");
30        seq.randomize();
31    endfunction
```

```
33    virtual task run_phase(uvm_phase phase);
34        phase.raise_objection(this);
35        apply_reset();
36        seq.start(e0.a0.s0);
37        #200;
38        phase.drop_objection(this);
39    endtask
40
41    virtual task apply_reset();
42        vif.rstn <= 0;
43        vif.in <= 0;
44        repeat(5) @ (posedge vif.clk);
45        vif.rstn <= 1;
46        repeat(10) @ (posedge vif.clk);
47    endtask
48 endclass
49
50 class test_1011 extends base_test;
51     `uvm_component_utils(test_1011)
52     function new(string name="test_1011", uvm_component parent=null);
53         super.new(name, parent);
54     endfunction
55
56     virtual function void build_phase(uvm_phase phase);
57         pattern = 4'b1011;
58         super.build_phase(phase);
59         seq.randomize() with { num inside {[300:500]}; };
60     endfunction
61 endclass
```

INTERFACE

```
1 // The interface allows verification components to access DUT signals
2 // using a virtual interface handle
3 interface des_if (input bit clk);
4     logic rstn;
5     logic in;
6     logic out;
7
8     clocking cb @(posedge clk);
9         default input #1step output #3ns;
10         input out;
11         output in;
12     endclocking
13 endinterface
```

TEST BENCH

```
1 module tb;
2   reg clk;
3
4   always #10 clk =~ clk;
5   des_if _if (clk);
6
7   det_1011 u0    ( .clk(clk),
8                   .rstn(_if.rstn),
9                   .in(_if.in),
10                  .out(_if.out));
11
12
13   initial begin
14     clk <= 0;
15     uvm_config_db#(virtual des_if)::set(null, "uvm_test_top", "des_vif", _if);
16     run_test("test_1011");
17   end
18 endmodule
```